
How to record i-8017H ’s Ch.1 to Ch.4 voltage input in a user allocated RAM 
memory in the Wincon-8xx7 ? The sampling time is one record every 0.01 
second. The record period is 1 to 10 minutes. Then PC can download this 
record and display it as a trend curve diagram by M.S. Excel.

The “Whmi_13.pia” can run in Wincon-8xx7 / 8xx6  with driver version of 3.36 or later version. 

New drive: http://www.icpdas.com/products/PAC/i-8000/isagraf-link1.htm
“whim_13.pia” resides at W-8xx7 CD-ROM:\napdos\isagraf\wincon\demo\  or
ftp://ftp.icpdas.com/pub/cd/wincon_isagraf/napdos/isagraf/wincon/demo/  or
www.icpdas.com – FAQ - Software – ISaGRAF – FAQ057

VB6 - “Demo_5” code at

W-8xx7 CD-ROM:\napdos\isagraf\wincon\vb6_demo_pc\  or
ftp://ftp.icpdas.com/pub/cd/wincon_isagraf/napdos/isagraf/wincon/vb6_demo_pc/  or
www.icpdas.com – FAQ - Software – ISaGRAF – FAQ057

If using Web HMI in this demo, the Web HMI codes resides at below location. (Please refer to 
Chapter 3, 4 and 5 in the “Wincon ISaGRAF Getting Started manual” or 
W-8xx7 CD-ROM:\napdos\isagraf\wincon\english_manu\ “getting_started_w8337.pdf”)

W-8xx7 CD-ROM:\napdos\isagraf\wincon\WebHMI_Demo\  or
ftp://ftp.icpdas.com/pub/cd/wincon_isagraf/napdos/isagraf/wincon/webhmi_demo/  or
www.icpdas.com – FAQ - Software – ISaGRAF – FAQ057

If new c-function of Msg_F , Msg_N , ARY_F_R and AFY_F_W doesn’t exist in ISaGRAF in 
your PC, please visit http://www.icpdas.com/products/PAC/i-8000/isagraf.htm to download the 
“ICP DAS utilities For ISaGRAF” . Then run “setup.exe” inside it to re-install all new ISaGRAF c-
function & I/O boards definition to your ISaGRAF workbench.

We use “Variable array” in this demo program, please refer to Chapter 2.6 or FAQ-039 for more 
information about “Variable array” .

How to test this demo ?

The following steps is only for using Web HMI as Human-Machine-Interface. If you are using VB 
6.0 – “Demo_5” as HMI , please run it (Demo_5.exe) in your PC and only do procedure listed in 
step 1 , step 3 and step 6 .

1. Please plug one i-8024 in W-8xx7 ’s Slot 2, one i-8017H in Slot 3. Then connect Ch1. to Ch.4 
voltage output of i-8024 to Ch1. to Ch.4 of i-8017H. Then power up Wincon, Check “Enable 
Web HMI” option as below.  For demo purpose, please don’t check “Enable Account Security”

2. Copy all files of Web HMI ‘s Demo_13 to Wincon ‘s  \CompactFlash\Temp\HTTP\WebHMI\ 
folder by ftp utility (For example, run ftp://10.0.0.103 in Internet Explorer)

Web HMI codes resides at 
W-8xx7 CD-ROM:\napdos\isagraf\wincon\WebHMI_Demo\  “whim_13”  folder  or
ftp://ftp.icpdas.com/pub/cd/wincon_isagraf/napdos/isagraf/wincon/webhmi_demo/ 

Ther are 7 Files plus 2 folder should be copied to Wincon ‘s 
\CompactFlash\Temp\HTTP\WebHMI\ 

Main.htm , menu.htm , index.htm , login.htm , main.dll , login.dll , whim_filter.dll
“img”  & “msg” folder

3. Download ISaGRAF project “whim_13” to W-8xx7. (If using Web HMI as HMI, please finish 
procedure listed in step 2 first, then do step 3 )

The “Whmi_13.pia” can only run in Wincon-8xx7/8xx6  (not in I-8xx7). It resides at  
W-8xx7 CD-ROM:\napdos\isagraf\wincon\demo\   or
ftp://ftp.icpdas.com/pub/cd/wincon_isagraf/napdos/isagraf/wincon/demo/ 

4. PC run Internet Explorer (I.E. version shoulde be 5.0 or later version). Enter W-8xx7 IP. If 
connecting well, click on “Enter”

5. Then please enter proper “Interval” value. Unit is 0.001 second (1 milli-second). For example, 
if enter 10, it means to store one record every 10 ms.  The ”Period” is the time period to record. 
Unit is minute.   Then please click on “Go” to start recording.  W-8xx7 will then output different 
voltage in i-8024 Ch.1 to Ch.4 . If user has finished procedure listed  in step 1 – “connect Ch1. 
to Ch.4 voltage output to Ch1. to Ch.4 of i-8017H” , the i-8017H Ch1. to Ch.4 ‘s voltage input 
will also change during this period. And they will be recorded.

During the recording period, the “Current record number” value will count up. If it reaches the 
value of  “Total record number” , it means recording is finished. Then the ISaGRAF program will 
store these records to a RAM file automatically. You can see the progress in “Saving state” . If 
all done, please click on “Download record File” to download this record file to your PC.

6. Then please open this record file - “curver1.js” on M.S. Excel.

Please click on the first data at the left-top position. Then press and hold in “Shift” , and at the 
same time press “Ctrl” – “End” . You will see all data been selected.

Then click on   

Please select the correct diagram on the left-hand side. And check the left-top type on the right-
hand side. Then go Next .

By the procedure, you will get the trend curve as the below window. You can modify its size, or 
check at any trend line. If you move your mouse to point at some position at the trend line, the 
related data is shown.

Please save this trend curve diagram as a “Microsoft Office Excel (*.xls)” format. Then at any 
later time, you can open it to display the trend curve directly.

ISaGRAF project architecture:

Variables : 
Name Type Attribute Description
Go1 Boolean Internl Set as True to start, addr defined as 21 (Hex. is 15)
Stop1 Boolean Internal Set as True to stop, addr defined as 22 (Hex. is 16)
TMP Boolean Internal Internal use
INIT Boolean Internal Init as True
Save_file1 Boolean Internal The IsaGRAF program will set this value to True to store 

records to a RAM Disk File
MUM_CH Integer Constant How many chanels in i-8017H to record ?

We use 4 channels in this demo (Ch.1  to 4)
File1 Integer Internal File ID
STEP1 Integer Internal Recording state. 0:No action , 1:recording , 2:finished
Period1 Integer Internal How long to record ? unit is minute, addr as 3
Interval1 Integer Internal How long to save a record ? unit is ms, addr as 1
Total_record1 Integer Internal How many records in this recording action ? This value 

is calculated by the IsaGRAF program automatically. 
addr declared as 5

Record_cnt1 Integer Internal Current finished record count. addr declared as 7
ii  & ii2 Integer Internal To use in “for” loops
i8017H[0..7] Integer Input Variable array, Dim as 8. link to i-8017H ‘s Ch1 to Ch. 8
Volt1[0..7] REAL Internal Variable array, Dim declared as 8. 

The voltage value converted from “i8017H[0..7]”
i8024[0..3] Integer Output Variable array, Dim declared as 4. link to i-8024 ‘s Ch1 

to  Ch. 4
Save_cnt1 Integer Internal Current saving record amount in the RAM disk File, addr 

declared as 9
TMP_v Integer Internal Internal use
T1 Timer Internal For counting time
T1_next Timer Internal The time to get and save next record
T1_Interval Timer Internal The interval time between two record
File_name1 Message Internal File name, Len is 64, init as   \Web_Data1\curve1.js

Web HMI support only RAM Disk File in   \Web_Data1\ , 
If the file is in CompactFlash File, Web HMI support only 
in \CompactFlash\Temp\HTTP\Data\    (Please refer to 
Chapter 11.2  - Whmi_08  demo)

Msg1 Message Internal Operation state message, Len is 255, init as  “No Action 
now”, addr  as 41 (Hex. is 29)

Str1 Message Intenal Len is 255, internal use

IO connection:

ST program – Sim_out

------------------------------------------------------------------------------------------------------------------------
(* Output i-8024 ‘s Ch1  to Ch4 as different voltage curve *)

(*   2 * Pi * T1 / 60000  =  T1 * 1.047197E-4    *)
(*   2 * Pi * T1 / 120000 = T1 *  5.235985E-5 *)

i8024[0]  :=  ANA(  sin( REAL(T1) *  1.047197E-4 ) * 3276.8  )  ;   
i8024[1]  :=  ANA(  cos( REAL(T1) *  5.235985E-5  ) * 3276.8  ) ;  
i8024[2]  :=  ANA(  sin( REAL(T1) *  1.047197E-4 ) * 6553.6  ) ;   
i8024[3]  :=  ANA(  cos( REAL(T1) *  5.235985E-5  ) * 6553.6  ) ;  
------------------------------------------------------------------------------------------------------------------------

ST program – ST1
------------------------------------------------------------------------------------------------------------------------

(* W-8xx7 can have max. speed of 100Hz to record data ( minimum sample interval is 10 ms) *)
(* This example assume max. 8-Ch. , so 1 second will record 100 x 8 REAL value *)
(*   1 minute will record  100 x 8 x 60 = 48,000 REAL value *)
(* If period is set as 10 minute, we need 48,000 x 10 = 480,000 REAL value memory = 480,000 
x 4 = 1,920,000 bytes *)

if  INIT  then
  INIT := False ;

  (* Allocate 500,000 integer (or 32-bit REAL) space to store records up to 10 minutes. total 
500,000 x 4 = 2,000,000 bytes  , W-8xx7 support only No.1 Arcreate( ) up to 3,000,000 integer 
space, that is 12,000,000 bytes . The first parameter in ARcreate( ) should be 1, it doesn't 
support 1st parameter as 0, 2, 3, ... , 15  *)
  (* Arcreate( ) can be called only once in the ISaGRAF program *)
  TMP_v := ARcreate(1, 500000) ;    
  if  TMP_v <> 1  then
     Msg1 := 'Parameter error or can not allocate memory by ARcreate( ) function!' ;
  end_if  ;

  TMP := PLC_mode(-1) ;     (* Set W-8xx7 ISaGRAF driver running at fatest mode *)
end_if ;

(* If stop command is gived *)
if  Stop1  then
  Stop1 := False ;
  STEP1 := 0 ;  (*  0: no action *)
  TStop(T1)  ;  (* stop T1 *)
  T1 := T#0s ;
  Msg1 := 'User stop recording !' ;

  save_cnt1 := 0 ;  
end_if  ;

(* If start command is gived *)
if  Go1  then

   Go1 := False ;

   (* STEP1 :  0: no action , 1: recording , 2: recond finished *)
   if  STEP1=1  then
   

       (* It is still recording now *)
       Msg1 := 'It is still recording now ... Please wait' ;

   else

       (* Check interval valid or not *)
       (* we assume 10 to 10000 ms is valid in this example *)
       (* If your average PLC scan time is larger, for example, near 10 ms, 
            Please use Interval larger than 10 ms. Or the record time won't be correct *) 
       if  (Interval1 < 10)  or  (Interval1 > 10000)  then
         Msg1 := 'Wrong Interval value, it should be in 10 to 10000 milli-second !' ;
       

       (* Check period valid or not *)
       (* we assume 1 to 10 minute is valid in this example *)
       elsif ( Period1 < 1)  or  (Period1 > 10)  then
         Msg1 := 'Wrong Period value, it should be in 1 to 10 minute !' ;

       else

         (* parameter is correct, start recording *)

         total_record1 := (Period1 * 60000) / Interval1 ;  (* calculate total record number *)
         record_cnt1 := 0 ;  (* reset current record count as 0 *)
         STEP1 := 1 ;  (* set step as 1:recording *)
         Msg1 := 'Recording now ... Please wait'  ;

         (* start ticking T1 from 0 second *)
         T1 := T#0s ;
         T1_Interval := TMR(Interval1) ;
         T1_next := T1 + T1_Interval ;
         TStart(T1) ;   (* ticking now *)
         save_cnt1 := 0 ;  
       end_if ;

   end_if ;
end_if ;

(* in reconrding state *)
if  STEP1 = 1  then

  (* store one record *)
  if  T1 >= T1_next  then

    (* Re-calculate next T1 *)
    T1_next := T1_next + T1_Interval ;

    (* T1 will be overflow after T#23h59m59s999ms, so reset it at T#20h *)
    if   T1 >= T#20h  then
      T1 := T#0s ;
      T1_next := T1 + T1_Interval ;
    end_if ;

    (* record data *)
    for  ii  :=  0  to  NUM_CH-1  do  
       Volt1[ii] := Real(i8017H[ii]) * 0.000305176 ;    (* convert to voltage *)

(* using Real_int( ) to map REAL value to become integer value & then store it by ARwrite( ) *)
       TMP_v := ARwrite( 1 ,  NUM_CH * record_cnt1 + ii , Real_int(Volt1[ii]) ) ;

       (* check if ARwrite( ) correct *)
       if TMP_v <> 1  then
          Msg1 := 'Can not operate ARwrite( ) !' ;
          STEP1 := 0 ;    (*  0: no action *)
          TStop(T1)  ;    (* stop T1 *)
          T1 := T#0s ;
       end_if ;

    end_for ;
        

    (* Check if record number reach the end *)
    record_cnt1 := record_cnt1+1 ;      (* current record count plus 1 *)
    if  (record_cnt1 >= total_record1)   then
 

 (* record is finished, prepare to save records to a RAM disk file in serval separate PLC scans *)
      STEP1 := 0 ;   (* set step as 0 at the beginning of saving *)
      Tstop(T1) ;
      T1 := T#0s ;

      (* Create a new file *)
      File1 := F_creat(File_name1) ;
      if   File1 = 0  then
         (* Can not create file *)
         Msg1 := 'Create File ' + 'File_nam1 Error !!!' ;
      else

        (* Because saving lots of data to file take lots of PLC scan time , we are not going to 
            save all data in a single PLC scan. We will save it in serval separate PLC scans *)

        Msg1 := ' Please wait ...   Saving data to file : ' + File_name1 + ' ...' ;
        save_file1 := True ;  (* set as True to start saving RAM disk file *)
        save_cnt1 := 0 ;   (* from 0 to total_record1-1 *)
      end_if ;

    end_if ;
  end_if ;
end_if ;

(* Because saving lots of data to file take lots of PLC scan time , we are not going to 
      save all data in a single PLC scan. We will save it in serval separate PLC scans *)

(* save records to a RAM disk file in serval separate PLC scans *)
if  save_file1  then
   for  ii2 := 0  to 50  do      (* we limit one PLC scan can save max. 50 records *)
      if  save_cnt1 < total_record1  then
         str1 := '' ;     (* init str1 as empty string *)   
         for  ii  :=  0  to  NUM_CH - 1  do  

(* delimiter is <TAB> character *)
            str1 := str1 + Rea_Str2( Int_real(ARread(1, NUM_CH * save_cnt1 + ii)) , 3  ) + '$09' ; 
         end_for ;
         str1 := str1 + '$0D$0A'  ;  (* add <CR> <LF> at the end of each row *)
         TMP := F_writ_s(File1 , str1) ;
         save_cnt1 := save_cnt1 + 1 ;
      else
         (* saving is finished *)
         save_file1 := False ;   
         TMP := F_close(File1) ;      (* Close file *) 
         STEP1 := 2 ;                        (* 2: recond finished *)
         Msg1 := 'Record is finished ! You may download the record file to your PC now !' ;
      end_if ;
   end_for ;
end_if ;  
  ------------------------------------------------------------------------------------------------------------------------

